Exercise 12.3.2

Scattering by a conducting disk

Exercise 12.3.2. Consider high-energy scattering of electromagnetic plane waves (initial direction $\hat{k}_0$, initial polarization vector $\hat{\epsilon}_0$) off of a perfect conductor in the shape of a flat disk of radius a. Consider only the special case of $\hat{k}_0$ perpendicular to the plane of the disk, as shown in Figure 12.6:

(a) Show that the *illuminated* side unpolarized differential cross section is ($\hat{k}_0 \cdot \hat{k} = \cos\theta$)

$$\left.\frac{d\sigma}{d\Omega}\right|_{\text{ill}} = \frac{a^2 J_1^2(ka\sin\theta)}{\sin^2\theta}\sin^4(\theta/2),$$

where $J_1(x)$ is an integer Bessel function.

(b) Plot this for increasingly large values of $ka \gg 1$. Notice the cross section peaks in the backward direction, as one would expect.



Fig. 12.6 Reference figure for Exercise 12.3.2.

Part (a)

Calculate the polarized cross section, starting from the surface integral over the illuminated side of the disk, equation (12.68)

▪Hint 1▪
Draw a picture of the problem.

▪Hint 2▪
Write eq. (12.68). Let the z-axis be along $\hat{k}_0$.
So, $\hat{k}_0 = k\,\hat{e}_z$, $\hat{n}' = -\hat{e}_z$. Rewrite eq (12.68) with these substitutions.

$$\hat{\epsilon}_f \cdot \vec{f}_{\text{ill}} = \frac{k}{4\pi i}\int_{\text{ill}} da' \exp\left\{-i(\vec{k}-\vec{k}_0)\cdot\vec{x}'\right.$$

$$\hat{\epsilon}_f \cdot [\,(\hat{k} - \hat{k}_0)\times(\hat{n}' \times \hat{\epsilon}_0) - \hat{k}_0(\hat{n}' \cdot \hat{\epsilon}_0)\,]$$

Let the z-axis be along $\vec{k}_0$;

$\vec{k}_0 = k\, e_z = k\,\{0, 0, 1\}$

$\epsilon_0^{[1]} = e_x = \{1,0,0\}$ and $\epsilon_0^{[2]} = e_y = \{0,1,0\}$

$\hat{n}' = -\hat{e}_z = (-1)\{0, 0, 1\}$

$\vec{k} = k\,\{\sin\theta\cos\phi, \sin\theta\sin\phi, \cos\theta\}$

$\epsilon_f^{[1]} = \{-\sin\phi, \cos\phi, 0\}$

$\epsilon_f^{[2]} = \{-\cos\theta\cos\phi, -\cos\theta\sin\phi, \sin\theta\}$

▪Hint 3▪
Rewrite eq (12.68) in this form :
prefactor ▫ integral ▫ Q
where Q does not depend on the integration variable.

$$\hat{\epsilon}_f \,.\, \vec{f}_{\rm ill} = \frac{k}{4\pi i} \int_{\rm ill} da' \;\exp\{-i(\vec{k}-\vec{k}_0)\,.\,\vec{x}'\}$$

$$\hat{\epsilon}_f \,.\, [\,(\hat{k} - \hat{e}_z) \times (\,\hat{\epsilon}_0 \times \hat{e}_z)\,]$$

The polarized cross section is

$$\left(\frac{d\sigma}{d\Omega}\right)_{f,i} = |\,\hat{\epsilon}_f \,.\, \vec{f}_{\rm ill}\,|^2$$

$$= \frac{k^2}{(4\pi)^2} \;\square\; |\text{ integral }|^2 \;\square\; |\,\hat{\epsilon}_f \,.\, \vec{Q}\,|^2$$

▪Hint 4▪
You have $\vec{k}-\vec{k}_0$ = k { sinθ cosϕ, sinθ sinϕ, cosθ - 1} and $\hat{x}'$ = ρ' { cosϕ', sinϕ', 0 } where ρ' and ϕ' are plane polar coordinates on the illuminated side of the disk. Do the integral over ϕ'. Do the integral over ρ'.

Calculate the surface integral over the illuminated side of the disk.

$$SI = \int_{\rm ill} da' \;\exp\{-i(\vec{k}-\vec{k}_0).\,\vec{x}'\}$$

Using plane polar coordinates (ρ' and ϕ').

$$SI = \int_0^a \rho'\, d\rho' \int_0^{2\pi} d\phi' \;\exp\{-i\,\vec{q}\,.\,\vec{x}'\}$$

where

$$\vec{q} = \vec{k} - \vec{k_0} = k\,\{\sin\theta\cos\phi\,,\, \sin\theta\sin\phi\,,\, \cos\theta - 1\}$$

and

$$\vec{x}' = \rho'\,\{\cos\phi'\,,\, \sin\phi'\,,\, 0\}.$$

$$\vec{q}\,.\,\vec{x}' = k\rho' \sin\theta \cos(\phi - \phi')$$

$$SI = \int_0^a \rho'\, d\rho' \int_0^{2\pi} d\phi' \;\exp\{-i\,\lambda\cos(\phi-\phi')\}$$

where $\lambda = k\rho' \sin\theta$

$$= 2\pi \int_0^a \rho'\, d\rho' \;\; J_0(\lambda)$$

$$= 2\pi \int_0^a \rho'\, d\rho' \;\; J_0(k\rho'\sin\theta);$$

now let $x = k\rho'\sin\theta$

$$= 2\pi \int_0^{ka\sin\theta} x\, dx \;\; J_0(x)\,/\,(k\sin\theta)^2$$

$$x\, J_0(x) = \frac{d}{dx}\,(x J_1(x))$$

$$= 2\pi\, ka \sin\theta\; J_1(ka\sin\theta)\,/\,(k\sin\theta)^2$$

$$= 2\pi \frac{a}{k} \frac{J1(ka \sin\theta)}{\sin\theta}$$

▪Hint 5▪ The unpolarized cross section is

$|\text{prefactor}|^2 \cdot |\text{integral}|^2 \cdot 1/2 \sum_{i=1}^{2} \sum_{i=1}^{2} Q^2$

Calculate the polarization sum, for these plane polarizations:

$\vec{k}_0 = k\{0,0,1\}$; $\hat{\epsilon}_0(1) = \{1, 0, 0\}$; $\hat{\epsilon}_0(2) = \{0, 1, 0\}$;

$\vec{k} = k\{ s\theta\, c\phi ,\, s\theta\, s\phi ,\, c\theta \}$;

$\hat{\epsilon}_f(1)=\{-s\phi, c\phi, 0\}$; $\hat{\epsilon}_f(2)=\{-c\theta\, c\phi ,-c\theta\, s\phi, s\theta\}$;

Use Mathematica to calculate the polarization sum

$$= \frac{1}{2} \sum_{i,j=1}^{2} |\, \hat{\epsilon}_f \cdot \vec{Q}\, |^2 .$$

In[56]:=
```
Remove["Global`*"]
ez = {0, 0, 1};
k0 = k*ez;
ϵ0[1] = {1, 0, 0};
ϵ0[2] = {0, 1, 0};
np = (-1) * ez;
ku = { sθ * cϕ , sθ * sϕ , cθ };
ϵf[1] = { -sϕ, cϕ, 0 };
ϵf[2] = { -cθ * cϕ, -cθ * sϕ, sθ };
checks = {ku.ϵf[1], ku.ϵf[2], ϵf[1].ϵf[2]} /. {sϕ^2 -> 1 - cϕ^2};
checks // Expand
```

Out[66]= {0, 0, 0}

In[67]:=
```
PS = (1/2) * Sum[Sum[
      Power[Dot[ϵf[j], Cross[ku - ez, Cross[ϵ0[i], ez]]], 2],
      {j, 1, 2}], {i, 1, 2}];
PS = PS /. {sϕ → Sqrt[1 - cϕ^2]};
PS = PS /. {sθ → Sqrt[1 - cθ^2]};
PS = PS // Expand // Simplify
```

Out[70]= $(-1 + c\theta)^2$

Result,

$$PS = (1 - \cos\theta)^2 = 4 \sin^4(\theta/2)$$

The unpolarized cross section

$$\frac{d\sigma}{d\Omega} = \frac{1}{2} \sum_{i,j=1}^{2} \left(\frac{d\sigma}{d\Omega}\right)_{f,i}$$

$$\frac{d\sigma}{d\Omega} = \frac{k^2}{(4\pi)^2} \left| 2\pi \frac{a}{k} \frac{J1(ka\sin\theta)}{\sin\theta} \right|^2 [4 \sin^4(\theta/2)]$$

$$\frac{d\sigma}{d\Omega} = a^2 \frac{J1^2(ka\sin\theta)}{\sin^2\theta} \sin^4\left(\frac{\theta}{2}\right)$$

Part (b)

Plots of dσ/dΩ versus θ for increasing values of ka.

Use a log scale for ka from 1 to 50.

In[72]:=
```
dσ[ka_, θ_] = (BesselJ[1, ka*Sin[θ]] / Sin[θ]) ^2*
   Sin[θ / 2] ^4;
```

In[73]:=
```
LogPlot[{dσ[1, θ], dσ[5, θ]}, {θ, 0, Pi},
 PlotStyle → {{Thickness[0.005], Blue},
   {Thickness[0.005], Red}},
 BaseStyle → "Text",
 Frame → True, ImageSize → Large]
```

Out[73]=



In[74]:=
```
LogPlot[{dσ[10, θ], dσ[50, θ]}, {θ, 0, Pi},
 PlotStyle → {{Thickness[0.005], Blue},
   {Thickness[0.005], Red}},
 BaseStyle → "Text",
 Frame → True, ImageSize → Large]
```

Out[74]=



Part (c)

Calculate $\sigma = \int \frac{d\sigma}{d\Omega}\, d\Omega$

In[75]:=
```
Plot[Sin[θ] * dσ[40, θ], {θ, 0, Pi}]
NIntegrate[Sin[θ] * dσ[40, θ], {θ, 0, Pi}]
```

Out[75]=



Out[76]= 0.494095

In[77]:=
```
list = {};
Do[x = 1.0 * j;
 s = NIntegrate[2 π * Sin[θ] * dσ[x, θ], {θ, 0, Pi}];
 list = Join[list, {{x, s}}],
 {j, 1, 50, 0.1}]
list;
ListPlot[list, Joined → True,
 PlotStyle → Thickness[0.005],
 BaseStyle → "Text",
 Frame → True, ImageSize → Large]
```

Out[80]=



Limiting value as $\lambda \longrightarrow 0$.

$ka = 2\pi a/\lambda \longrightarrow \infty$

In[81]:=
```
x = 1000
NIntegrate[2 π * Sin[θ] * dσ[x, θ], {θ, 0, Pi}]
```

Out[81]= 1000

NIntegrate: NIntegrate failed to converge to prescribed accuracy after 9 recursive bisections in θ near {θ} = {2.92679}. NIntegrate obtained 3.140045011382462` and 0.0000972030349254103` for the integral and error estimates.

Out[82]= 3.14005

Evidently $\lim_{ka \to \infty} \sigma_{total} = \pi a^2$ = the geometrical cross section.